<commit_before>{ stdenv, cmake, fetchurl, gnumake, pkgconfig
, boost, gettext, tclap, wxGTK
, freeglut, glew, libXi, libXmu, mesa
, autopanosiftc, enblend-enfuse, exiv2, ilmbase, lensfun, libpng, libtiff
, openexr, panotools, perlPackages
}:

stdenv.mkDerivation rec {
  name = "hugin-2013.0.0";

  src = fetchurl {
    url = "mirror://sourceforge/hugin/${name}.tar.bz2";
    sha256 = "1mgbvg09xvf0zcm9jfv5lb65nd292l86ffa23yp4pzm6izaiwkj8";
  };

  NIX_CFLAGS_COMPILE = "-I${ilmbase}/include/OpenEXR";

  buildInputs = [ boost gettext tclap wxGTK
                  freeglut glew libXi libXmu mesa
                  exiv2 ilmbase lensfun libtiff libpng openexr panotools
                ];

  # disable installation of the python scripting interface
  cmakeFlags = [ "-DBUILD_HSI:BOOl=OFF" ];

  nativeBuildInputs = [ cmake pkgconfig ];

  enableParallelBuilding = true;

  # commandline tools needed by the hugin batch processor
  # you may have to tell hugin (in the preferences) where these binaries reside
  propagatedUserEnvPackages = [ autopanosiftc enblend-enfuse gnumake
                                perlPackages.ImageExifTool
                              ];

  postInstall = ''
    mkdir -p "$out/nix-support"
    echo $propagatedUserEnvPackages > $out/nix-support/propagated-user-env-packages
  '';

  meta = {
    homepage = http://hugin.sourceforge.net/;
    description = "Toolkit for stitching photographs and assembling panoramas, together with an easy to use graphical front end";
    license = stdenv.lib.licenses.gpl2Plus;
    maintainers = with stdenv.lib.maintainers; [viric];
    platforms = with stdenv.lib.platforms; linux;
  };
}<commit_msg>hugin: update 2013.0.0 -> 2015.0.0<commit_after>{ stdenv, cmake, fetchurl, gnumake, pkgconfig, makeWrapper
, boost, gettext, tclap, wxGTK
, freeglut, glew, libXi, libXmu, mesa
, autopanosiftc, enblend-enfuse, exiv2, fftw, ilmbase, lensfun, libpng, libtiff
, openexr, panotools, perlPackages, sqlite, vigra
}:

stdenv.mkDerivation rec {
  name = "hugin-2015.0.0";

  src = fetchurl {
    url = "mirror://sourceforge/hugin/${name}.tar.bz2";
    sha256 = "1gfblax9rxay8xskz5r8bips4nfh70vkyrb8ksgl6pg91c8krn9c";
  };

  NIX_CFLAGS_COMPILE = "-I${ilmbase}/include/OpenEXR";

  buildInputs = [ boost gettext tclap wxGTK
                  freeglut glew libXi libXmu mesa
                  exiv2 fftw ilmbase lensfun libtiff libpng openexr panotools
                  sqlite vigra
                  perlPackages.ImageExifTool makeWrapper
                ];

  # disable installation of the python scripting interface
  cmakeFlags = [ "-DBUILD_HSI:BOOl=OFF" ];

  nativeBuildInputs = [ cmake pkgconfig ];

  enableParallelBuilding = true;

  postInstall = ''
    for p in $out/bin/*; do
      wrapProgram "$p" \
        --suffix PATH : ${autopanosiftc}/bin \
        --suffix PATH : ${enblend-enfuse}/bin \
        --suffix PATH : ${gnumake}/bin \
        --suffix PATH : ${perlPackages.ImageExifTool}/bin
    done
  '';

  meta = {
    homepage = http://hugin.sourceforge.net/;
    description = "Toolkit for stitching photographs and assembling panoramas, together with an easy to use graphical front end";
    license = stdenv.lib.licenses.gpl2Plus;
    maintainers = with stdenv.lib.maintainers; [ <NAME> ];
    platforms = with stdenv.lib.platforms; linux;
  };
}
{'fraction_non_alphanumeric': 0.1036970243462579, 'fraction_numerical': 0.021039975954313193, 'mean_word_length': 3.0096385542168673, 'pattern_counts': {'":': 0, '<': 4, '<?xml version=': 0, '>': 6, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3097019', 'n_tokens_mistral': 1244, 'n_tokens_neox': 1140, 'n_words': 262}